
ABSTRACT
Two methods to analyzing survival data with clustered events are presented. The first method is a 
proportional hazards model which adopts a marginal approach with a working independence 
assumption. This model can be fitted by SAS PROC PHREG with the robust sandwich estimate 
option. The second method is a likelihood-based random effects (frailty) model. In the second 
model, the baseline hazard could be either a priori determined (e.g., Weibull) or approximated by 
piecewise constant counterpart. The estimation could be carried out by adaptive Gaussian 
quadrature method which is implemented in SAS PROC NLMIXED. The advantages,
disadvantages, and relevant situations for proper application of each model are demonstrated
using the published diabetic retinopathy data. 

Keywords: survival analysis, Cox's proportional-hazards model, survival function, frailty model, 
mixed models, Gaussian quadrature estimation, PROC PHREG, and PROC NLMIXED.

INTRODUCTION
Survival analysis is the phrase used to describe the analysis of data in the form of times from a 
well-defined “time origin” until the occurrence of some particular event or “end-point” such as 
death. In this paper we are interested in the analysis of multivariate failure time data. Such data 
arise when each study subject can potentially experience several events (for instance, multiple 
infections after surgery) or when there exists some natural or artificial clustering of subjects (for 
instance, a litter of mice) that induces dependence among the failure times of the same cluster. 
Data in the former situation are referred to as multiple events data, which include recurrent events 
data as a special case; data in the latter situation are referred to as clustered data. 
Two approaches are generally used in the analysis of correlated failure time data: a marginal 
model approach (Wei, Lin, and Weissfeld [1], hereafter WLW) and a frailty model approach [2]. In 
the marginal model, WLW adopts a pseudolikelihood approach with a working independence 
assumption. They estimated the covariate effect assuming independence among correlated 
failure times, while adjusting for the correlation by sandwich estimate in estimating the covariance 
matrix. Frailty model is a random effects proportional hazards model. The random effects could 
be used to capture the relation of the correlated observations in the clustered or recurrent events 
data.
There are three SAS procedures available for analyzing survival data: LIFEREG, LIFETEST and 
PHREG. PROC LIFEREG is a parametric regression procedure to model the distribution of 
survival time with a set of concomitant variables [3]. PROC LIFETEST is a nonparametric 
procedure to estimate the survivor function, compare the underlying survival curves of two or 
more samples [4]. PROC PHREG is a semi-parametric procedure that fits the Cox proportional 
hazards model [5]. These procedures can not be used directly in modeling correlated survival 
data. However, with the robust sandwich estimate option, PROC PHREG can be used to perform 
clustered data analysis or recurrent data analysis, adopting a GEE-like marginal approach. This 
procedure will be illustrated under Model 1.

In SAS, the estimation in frailty model could be carried out in PROC NLMIXED. The NLMIXED 
procedure fits nonlinear random effects models using MLE. Both fixed and random effects can be 

included the model nonlinearly [6, 7]. These models have a wide variety of applications, e.g.,
pharmacokinetics and overdispersed binomial data. 

PROC NLMIXED fits nonlinear mixed models by maximizing an approximation to the likelihood 
integrated over the random effects [5, 6]. Different integral approximations are available, the 
principal ones being adaptive Gaussian quadrature and a first-order Taylor series approximation. 
A variety of alternative optimization techniques are available to carry out the maximization; the 
default is a dual quasi-Newton algorithm which updates the Cholesky factor of an approximate 
Hessian to achieve convergence [6]. 

Successful convergence of the optimization problem results in parameter estimates along with 
their approximate standard errors based on the second derivative matrix of the likelihood function. 
PROC NLMIXED enables the users to apply the estimated model to construct predictions of 
arbitrary functions by using empirical Bayes estimates of the random effects. The users can also 
estimate the arbitrary functions of the nonrandom parameters, and PROC NLMIXED computes 
their approximate standard errors by using the delta method. 

SAS PROC NLMIXED enables the user to specify a conditional distribution for the data (given the 
random effects) having either a standard form (normal, binomial, Poisson) or a general 
distribution that the user can code using SAS programming statements. The latter feature makes 
it possible to fit a frailty model with either an a priori (e.g., Weibull) parametric baseline hazard 
function or a piecewise constant one. We will illustrate these two models under models 2A and 
2B, respectively.

SURVIVAL DATA USED
The three models described in this paper are applied to analyze the same dataset to be 
consistent: the Diabetic Retinopathy Study (DRS) dataset which is available online at
http://www.mayo.edu/hsr/people/therneau/book/data/diabetes.html. Diabetic retinopathy is the
most common and most serious eye complication of diabetes, which may lead to poor vision or
even blindness. The DRS was started in 1971 to investigate the efficacy of laser 
photocoagulation in delaying the onset of severe vision loss. This study enrolled 1742 patients. 
One eye of each patient was randomly chosen for treatment and the other eye served as an 
untreated control. Huster et al. [7], Therneau and Grambsch [8] both analyzed a small part of the 
total data, which is a 50 per cent sample of the high-risk patients (N =197).

Each patient had one eye randomized to laser treatment (laser type 1=xenon, 2=argon) and the 
other eye received no treatment. For each eye, the event of interest was the time from initiation of 
treatment to the time when visual acuity dropped below 5/200 two visits in a row (defined as 
"blindness"). Thus there is a built-in lag time of approximately 6 months (visits were every 3 
months). Survival times in this dataset are therefore the actual time to blindness in months, minus 
the minimum possible time to event (6.5 months). Censoring could be caused by death, dropout, 
or end of the study. Two covariates are included in the analysis. The Treat covariate indicates an 
eye received laser treatment. Treat=1 indicates right eye and Treat=2 indicates left eye. The JR 
covariate indicates the type of diabetes with JR=1 for juvenile (age at diagnosis < 20) and JR=2 
for adult. The outcome for the treated eye was compared to the untreated eye for risk groups 6-
12 and Status=0 shows censored data and Status=1 shows blindness. 

DEMONSTRATION OF MODELS FOR ANALYSIS OF SURVIVAL DATA 

MODEL 1: ANALYSIS OF CLUSTERED DATA USING PROC PHREG

1.1 MARGINAL COX MODELS FOR MULTIPLE EVENTS DATA
Suppose there are i =1,2, . . . ,n clusters (centers, families, litters, etc.), where each cluster has

j =1,2, . . . ,J  members. We record the follow-up time ijX for each member, which is the minimum of the 

failure time ijD and the non-informative censoring time ijC . Denote by )( ijijij CDI  the event 

indicator, where I(.) is the indicator function. For each member there is a covariate vector )(tZij for fixed 

effects at time t.

The marginal Cox model for the jth event and the ith cluster is given by 

niJjeZt tZ
jijj

ijj ,...,1;,...,1,);( )(
0  

and the jth event-specific partial likelihood is

Where )(0 tj is an arbitrary baseline hazard function for the jth event and j is the (event-

specific) column vector of regression coefficients for jth event. WLW estimates J ,...,1 by the 

maximum partial likelihood estimates J ˆ,...,1̂ , respectively, and uses a robust sandwich 

covariance matrix estimate for )ˆ,...,ˆ( 1  J to account for the dependence of the multiple failure 
times. 

1.2 IMPLEMENTATION IN SAS PROC PHREG
By using a properly prepared input data set, the user can estimate the regression parameters for 
all the marginal Cox models and compute the robust sandwich covariance estimates in one 
PROC PHREG invocation. The PHREG procedure uses ODS Graphics to create graphs as part 
of its output. For example, the ASSESS statement uses a graphical method that uses ODS 
Graphics to check the model adequacy of the model. PROC PHREG (Version 9.2) also offers a 
number of important enhanced features. The most noticeable features are the CLASS statement 
for specifying categorical variables; the CONTRAST statement for estimating and testing linear 
contrasts; the BAYES statement for performing a Bayesian analysis; and the HAZARDRATIO 
statement for estimating customized hazard ratios.

A sketch of the SAS program is given in Code Box 1 in the Appendix. The option 
COVS(AGGREGATE) is specified in the PROC statement to obtain the robust sandwich estimate 
of the covariance matrix, and the score residuals used in computing the middle part of the
sandwich estimate are aggregated over identical ID values. Then, the model is defined including 
Treat covariate, JR covariate, and interaction term between Treat and JR. The TEST statements
can be included in the PROC PHREG code to test various linear hypotheses of the regression 
parameters based on the robust sandwich covariance matrix estimate.

This method is most appropriate when the main purpose is to estimate the marginal (population) 
covariate effects. In applications where the estimates of the random effects to capture the 
correlation between multivariate failure times, frailty models should be used which will be 
described in the following sections.   

MODEL 2A: ANALYSIS OF CLUSTERED DATA WITH KNOWN DISTRIBUTION USING PROC 
NLMIXED

2A.1 INTRODUCTION TO FRAILTY MODEL

Denote by iv  the unobserved random effect (frailty) for the ith center with a density (.)f , which 
is often assumed to follow a log-Gamma (hereafter referred to as ‘Gamma’) or normal distribution, 
although other distributions have been proposed, e.g. positive stable [9]. Note that here we put 
the frailty term in the exponential component. Conditional on iv , a ‘simple frailty model’ can be 
written as 

))(exp()()( iij
T

oij vtZthth  

where )(tho is the unspecified baseline hazard and  is the regression coefficient. The observed

data },...,1,,{ iijiji njTO  are i.i.d for distinct center i . The likelihood for iO is

ii

x

ij

n

j
ijiji dvvfdtthxhOL

iji
ij )(])(exp[)]([)(

01
  



 



The diabetic retinopathy data described earlier has a hazard function

)JATR(t)exp()( 210 iijijij vhth  

Using notation previously defined, the contribution for each eye to the log-likelihood is

)(xH
~

)JATRexp(-)](x
~

logJATR[l
~

0210i21ij ijiijijijijijij vh  

2A.2 HOW TO FIND THE RIGHT DISTRIBUTION
The best way to find the right distribution is to fit several frailty models with different distributions 
for baseline hazard. Then, the model with best loglikelihood can be determined. This approach 
and related code development is currently ongoing and the results will be presented later. 
However, for illustration purposes, the PROC LIFEREG can be used to determine the right 
distribution assuming only fixed effects. The plot of the cumulative baseline hazard estimates 
from LIFEREG is shown in Figure 1. The -2log likelihood statistics derived from four distributions 
(Lognormal, Gamma, Exponential, and Weibull) using SAS PROC LIFEREG in Table 1 show that 
the Gamma distribution is most suited for this data when the random or clustered effects are 
ignored. 

Fit Statistics -2 Log Likelihood

Lognormal 897.838

Gamma 897.752

Exponential 914.613

Weibull 905.481

However, in the presence of random effects, LIFEREG may not reveal the right distribution. The 
estimated CDF, a line representing the maximum likelihood fit, and point-wise parametric 
confidence bands are plotted in the Gamma probability Plot using SAS ODS graphics feature. 
(Please see Code Box 2). The SAS code for estimation of frailty model with Gamma baseline 
hazard is presented in Code Box 3.

Table 1: Comparison of Four Distributions

MODEL 2B: ANALYSIS OF CLUSTERED DATA WITH UNKNOWN DISTIBUTION USING 
PROC NLMIXED

Liu and Huang proposed a novel Gaussian quadrature estimation method for various frailty 
proportional hazards models [10]. They approximated the unspecified baseline hazard by a 
piecewise constant one, resulting in a parametric model that can be fitted conveniently by 
Gaussian quadrature tools in standard software such as SAS PROC NLMIXED. In this model,
the estimates for the piecewise constant baseline hazard function are available. Therefore, the 
user can construct cumulative baseline hazard function (and thus survival function) from these 
estimates.

They first applied this method to simple frailty models for correlated survival data (e.g. recurrent 
or clustered failure times), then to joint frailty models for correlated failure times with informative 
dropout or a dependent terminal event such as death. Simulation studies showed this method 
compared favorably with the well-received penalized partial likelihood method and the Monte 
Carlo EM (MCEM) method. A subset of their novel approach will be illustrated in Model 2B.

SIMPLE FRAILTY MODEL FOR DIABETIC RETINOPATHY DATA
A piecewise constant baseline hazard with 10 intervals is used to fit the data by Proc NLMIXED,
assuming normal frailty model. The follow-up of the observed event times is divided into 10 
intervals by every 10th quantiles. The baseline hazard is

The cumulative baseline hazard is

Figure 1: Gamma Probability Plot

A sketch of the SAS program is given in Code Box 4 and Code Box 5 in the Appendix. The SAS 
data set ‘diabetes’ is prepared so that each patient has multiple data lines, each data line 
corresponds to the record of the failure time of an eye. In Code Box 4, the first step involves
finding 10 knots using quantized intervals. Since the goal is to estimate the 10 knots, PROC 
UNIVARIATE is used. The next data step shows the estimate for each quantized interval. The 
duration of each quantization interval is calculated together with the indicator of event in each 
interval. Then, the frequency of blindness event in each interval is shown (status = 1 means that 
blindness event has occurred). 

In Code Box 5, the parametric model is fitted by Gaussian quadrature tools in PROC NLMIXED 
using normal distribution. For each record there are 10 variables kd , k=1,2, . . . , 10, for the length 

within intervals, and 10 variables ke , k=1,2, . . . , 10, for the indicator of the failure happening in 

intervals 1–10, respectively. 
0

~

h is denoted by ‘basehaz’ and 
0

~

H by ‘cumhaz’ in the SAS code. 

Each subject has a unique ‘ID’. The follow-up time is denoted by ‘T’ and the event indicator ij by 
‘status’. The random effect is shown by ‘nu’. With normal random effects, the Gaussian 
quadrature technique can be applied directly. However, some adaption is needed when the 
random effect has another distribution, e.g. Gamma. Nelson et al. (2007) presented a solution by 
making use of probability integral transformation (PIT). For ),1,0(~ Na ),1,0(~)( Uniforma
where (.)  is the standard normal cumulative distribution function (CDF). Likewise, the CDF of a 
non-normal random variable b is Uniform (0,1), i.e. ).1,0(~)( Uniformbf Denoting the inverse 

CDF of b by (.),1F ))((1 aFb  can be used to generate the non-normal random variable of 
interest. Liu and Yu (2008) proposed to use a likelihood reformulation method which can be 
applied to more complicated situations. Their method is often much faster in computation than 
Nelson et al.

The Bounds statement requires that the 10 knots (h1 ….. h10) and variance of the random effect 
(theta) must be non-negative. The baseline (basehaz) and cumulative hazard (cumhaz) are 
computed and the model is written with beta1 as the treat covariate, beta2 as the JR covariate, 
beta3 as the interaction between beta1 and beta2, and ‘nu’ as the random effect. Random effect 
has normal distribution with the mean zero and variance theta. Then, the log likelihood for the 
model for both censoring and non-censoring data is estimated. Finally, the status statement is 
used to indicate the blindness event and non-event.

This method can handle the unknown distributions for baseline hazard and generate the correct 
estimates for treatment effects and random effects. It has several advantages. First, piecewise 
constant baseline hazard gives a close approximation to unspecified baseline hazards. The 
resulting parametric model is then conveniently maximized by Gaussian quadrature tools as in 
PROC NLMIXED. This method yields satisfactory results as far as parameter estimation is 
concerned. Another approach is the Penalized Partial Likelihood (PPL), which was first proposed 
by McGilchrist and Aisbett [11] for frailty models. Compared with the PPL method, this method 
has a smaller bias and can provide the standard error estimate (for frailty variance) directly.
Furthermore, this method can be easily applied to the joint frailty models while such an extension 
for the PPL method is not currently available. Compared with the Monte Carlo EM (MCEM)
approach, an appealing feature of this method is the ease of implementation, which can greatly 
enhance the applications of joint frailty models in practice. Monte Carlo MCEM method is often 
highly computationally intensive and its implementation (programming) is quite difficult and has to 
be treated on a case-by-case basis. This method also reduces computational time substantially. 
Finally, this approach is adaptable and can be easily applied to many other settings such as non-
proportional hazards models. 

RESULTS AND OUTPUTS

Table 2 shows the estimates of covariate effects are close in the three models but slightly higher 
for the normal frailty model in Model 2A. The standard error estimates are close too. Treatment
and JR interaction are significant in the three methods. JR is not significant in all three models 
based on the 5% significance level. In model 1, there is no estimate for random effects. In 
comparing Model 2A and Model 2B, the random effect is significant in Model 2B but not 
significant in Model 2A. This may be caused by the fact that Gamma distribution still does not 
capture the correct shape of the baseline hazard. The -2log likelihood statistics derived from 
Model 2A is 1646.0 and from Model 2B is 1629.2. Also, Model 2A has AIC of 1658.0 and BIC of 
1677.7, while Model 2B has AIC of1657.2 and BIC of 1703.1. Therefore, Model 2A should be 
preferred by BIC while Model 2B should be preferred by AIC.

SUMMARY  
Model 1 is a marginal model which can be easily implemented. This model is a pseudolikelihood 
method which yields a consistent estimate of covariate effects. However, sandwich estimate 
should be used to obtain a consistent estimate of covariance matrix.

Model 2A is a random effects model which requires an assumption on the distribution for the 
baseline hazard. The consistency of the estimates depends on the correct assumption of such 
distribution. Therefore, the researcher should have a good prior knowledge of the baseline 
survival function.

Model 2B is a random effects model which does not require assuming a distribution for the 
baseline hazard. Rather it is approximated by a piecewise constant baseline hazard. So this 
method is more flexible than Model 2A. The performance of this method can be compared with 
that of Model 2A by AIC or BIC criteria.

Analysis of Maximum Likelihood Estimates

PROC PHREG Model 1 PROC NLMIXED Model 2A PROC NLMIXED Model 2B

Variable Par. Est. Std err P-value Par. Est. Std err P-value Par. Est. Std err P-value

Treat -1.27 0.24 <.0001 -1.59 0.30 <.0001 -1.51 0.29 <.0001

JR -0.34 0.20 0.08 -0.45 0.27 0.10 -0.41 0.26 0.12

Treat*JR 0.85 0.30 0.005 1.04 0.37 0.006 0.98 0.37 0.008

Random 
Effects 0.13 0.14 0.36 1.03 0.37 0.005

Table 2: Comparison of Outputs from Three Models

APPENDIX

Code Box 1

Code Box 2

Code Box 3

Code Box 4

proc phreg data=diabetes covs(aggregate);
      model Time*Status(0)=Treat JR Interaction;
      Interaction= Treat * JR;
      id ID;
      run;

Title "Random Frailty Model with Gamma ";
ods output ParameterEstimates=est;
   proc nlmixed data=five qpoints=30 noad;
       bounds gamma > 0;
       interaction=treat*JR;
       linp  = b0 - b1*(treat)-b2*(JR)-b3*(interaction)+ z;
       alpha = exp(-linp);
       G_t   = exp(-(alpha*time)**gamma);
       g     = gamma*alpha*((alpha*time)**(gamma-1))*G_t;
       ll = (status=1)*log(g) + (status=0)*log(G_t);
       model time ~ general(ll);
       random z ~ normal(0,exp(2*logsig)) subject=id out=EB;
       predict 1-G_t out=cdf;
   run;

Ods Graphics ON;;
PROC LIFEREG CODE:
proc lifereg data=diabetes;
title "Lifereg =&distribution ";
      class TREAT JR;
      model Time*Status(0)=treat jr/dist=&distribution;
probplot  / nodata; inset;
   run;
ODS Graphics OFF;

Code Box 4

proc univariate data=diabetes noprint;
var time;
output out=quant_d pctlpts=0 10 20 30 40 50 60 70 80 90 100 pctlpre=qd;
where status=1;
run;
* Merge data with the quantiles;
data four;
set diabetes;if _n_ =1 then set quant_d;
run;
                                                                                                                                        
* Calculate the duration in each quantile interval, together with the indicator of event in 
each interval;
data five;
set four;
array quant_d {11} qd0 qd10 qd20 qd30 qd40 qd50 qd60 qd70 qd80 qd90 qd100;

array d {10} d1-d10;

array e {10} e1-e10;

do i=1 to 10;

    d{i}=0;
end;

do i=1 to 10;

    e{i}=0;
end;
    do i=2 to 11; /* duration in each death event quantiles */
        if time<=quant_d{i} then do;
            e{i-1}=(status=1);        /* indicator of blind event in each interval */
            d{i-1}=time-quant_d{i-1};
            i=11;
        end;
        else d{i-1}=quant_d{i}-quant_d{i-1};
    end;

run;

Code Box 5

proc nlmixed data=five qpoints=5;
Title "Normal frailty with variance theta";
parms h1=1 h2=1 h3=1 h4=1 h5=1 h6=1 h7=1 h8=1 h9=1 h10=1 beta1=1 beta2=-1 theta=1;

bounds h1 h2 h3 h4 h5 h6 h7 h8 h9 h10   theta >= 0;

basehaz=h1*e1+h2*e2+h3*e3+h4*e4+h5*e5+h6*e6+h7*e7+h8*e8+h9*e9+h10*e10;

cumhaz=h1*d1+h2*d2+h3*d3+h4*d4+h5*d5+h6*d6+h7*d7+h8*d8+h9*d9+h10*d10;
interaction=TREAT *JR ;
mu= (beta1 * TREAT) + ( beta2 * JR ) + (beta3 *interaction )+ nu;

loglik0=-exp(mu) * cumhaz;

if status=0 then loglik=loglik0;                           /* log likelihood for censoring */
if status=1 then loglik= log(basehaz) + mu + loglik0;      /* for failure */

model Time ~ general(loglik);

random nu ~ normal(0, theta) subject=ID out=eb;

run;

